Problem:
Score : 400 points

Problem Statement

Given is a rooted tree with N vertices numbered 1 to N.
The root is Vertex 1, and the i-th edge (1 \leq i \leq N - 1) connects Vertex a_i and b_i.

Each of the vertices has a counter installed. Initially, the counters on all the vertices have the value 0.

Now, the following Q operations will be performed:

Operation j (1 \leq j \leq Q): Increment by x_j the counter on every vertex contained in the subtree rooted at Vertex p_j.

Find the value of the counter on each vertex after all operations.

Constraints

2 \leq N \leq 2 \times 10^5

1 \leq Q \leq 2 \times 10^5

1 \leq a_i < b_i \leq N

1 \leq p_j \leq N

1 \leq x_j \leq 10^4

The given graph is a tree.

All values in input are integers.

Input

Input is given from Standard Input in the following format:

N Q
a_1 b_1
:
a_{N-1} b_{N-1}
p_1 x_1
:
p_Q x_Q

Output

Print the values of the counters on Vertex 1, 2, \ldots, N after all operations, in this order, with spaces in between.

Sample Input 1

4 3
1 2
2 3
2 4
2 10
1 100
3 1

Sample Output 1

100 110 111 110

The tree in this input is as follows:

Each operation changes the values of the counters on the vertices as follows:

Operation 1: Increment by 10 the counter on every vertex contained in the subtree rooted at Vertex 2, that is, Vertex 2, 3, 4. The values of the counters on Vertex 1, 2, 3, 4 are now 0, 10, 10, 10, respectively.

Operation 2: Increment by 100 the counter on every vertex contained in the subtree rooted at Vertex 1, that is, Vertex 1, 2, 3, 4. The values of the counters on Vertex 1, 2, 3, 4 are now 100, 110, 110, 110, respectively.

Operation 3: Increment by 1 the counter on every vertex contained in the subtree rooted at Vertex 3, that is, Vertex 3. The values of the counters on Vertex 1, 2, 3, 4 are now 100, 110, 111, 110, respectively.

Sample Input 2

6 2
1 2
1 3
2 4
3 6
2 5
1 10
1 10

Sample Output 2

20 20 20 20 20 20

Solution in OCaml (4.02.3) (judge verdict: Accepted)

let scan_int () = Scanf.scanf " %d" (fun x -> x)
let print_array a = 
  let _ = Array.iter (Printf.printf "%d ") a in print_newline()

module Graph :
  sig
    type graph
    type edge = int * int * int
    val init : int -> graph
    val get_size : graph -> int
    val get_edge: graph -> int -> edge list
    val add_edge: graph -> int -> int -> int -> unit
    val add_biedge: graph -> int -> int -> int -> unit
    val dfs_tree: graph -> int -> int -> (edge -> unit) -> unit
    val dfs_rec: graph -> int -> bool array -> (edge -> unit) -> unit
    val dfs: graph -> (edge -> unit) -> unit
  end =
  struct
    type edge = int * int * int (* from, to, dist *)
    type graph = {v: int; e: (edge list) array}
    let init _n = {v = _n; e = Array.make _n []}
    let get_size g = g.v
    let get_edge g u = g.e.(u)
    let add_edge g u v d = g.e.(u) <- (u, v, d) :: g.e.(u)
    let add_biedge g u v d = let _ = add_edge g u v d in add_edge g v u d
    let rec dfs_tree g curr prev func =
      let transit curr prev edge =
        let _, next, _ = edge in
        if next = prev then ()
        else let _ = func edge in dfs_tree g next curr func in
      List.iter (transit curr prev) g.e.(curr)
    let rec dfs_rec g curr visited func =
      let _ = visited.(curr) <- true in
      let transit curr visited edge =
        let _, next, _ = edge in
        if visited.(next) then ()
        else let _ = func edge in dfs_rec g next visited func in
      List.iter (transit curr visited) g.e.(curr)
    let dfs g func = 
      let visited = Array.make g.v false in
      for i = 0 to g.v - 1 do (if visited.(i) = false then dfs_rec g i visited func) done
  end


let make_graph g n =
  for i = 0 to n-2 do
    let a = scan_int() and b = scan_int() in
    Graph.add_biedge g (a-1) (b-1) 1
  done

let input_query q value =
  for i = 0 to q-1 do
    let p = scan_int() and x = scan_int() in
    value.(p-1) <- value.(p-1) + x
  done

let propagate value edge = let u, v, d = edge in value.(v) <- value.(u) + value.(v)


let () =
  let n = scan_int() and q = scan_int() in
  let g = Graph.init n in
  let _ = make_graph g n in
  let value = Array.make n 0 in
  let _ = input_query q value in
  let _ = Graph.dfs_tree g 0 n (propagate value) in
  print_array value